

Student Expectations

To solve this assignment students are expected to avail themselves of references provided in class and on the Web site, such as for Verilog programming and synthesis examples, and to seek out any additional help and resources that might be needed. (Of course this doesn't mean asking someone else to solve it for you.) It is the students' responsibility to resolve frustrations and roadblocks quickly. (If you get stuck *just ask for help!*)

This assignment cannot be solved by blindly pasting together parts of past assignments. Solving the assignment is a multi-step learning process that takes effort, but one that also provides the satisfaction of progress and of developing skills and understanding.

Collaboration Rules

Each student is expected to complete his or her own assignment. It is okay to work with other students and to ask questions in order to get ideas on how to solve the problems or how to overcome some obstacle (be it a question of Verilog syntax, interpreting error messages, how a part of the problem might be solved, etc.) It is also acceptable to seek out digital design resources for help on Verilog, digital design, etc. It is okay to make use of AI LLM tools such as ChatGPT and Copilot to generate sample Verilog code. (Do not assume LLM output is correct. Treat LLM output the same way one might treat legal advice given by a lawyer character in a movie: it may sound impressive, but it can range from sage advice to utter nonsense.)

After availing oneself to these resources **each student is expected to be able to complete the assignment alone**. Test questions will be based on homework questions and **the assumed time needed to complete the question will be for a student who had solved the homework assignment on which it was based**.

Problems start on next page.

Problem 1: Appearing below is the base case from module `pmatch_mark` in the solution to Homework 3 Problem 2.

```
typedef enum logic [3:0]
{ Char_Blank = 0,      Char_Dot = 1,
  Char_Open = 2,       Char_Close = 3,
  Char_Open_Okay = 4, Char_Close_Okay = 5 } Char;

module pmatch_mark
#( int n = 5, wn = $clog2(n+1) )
( output logic [wn-1:0] left_out_n_unmat_close, right_out_n_unmat_open,
  output uwire [3:0] str_marked [0:n-1],
  input uwire [wn-1:0] left_in_n_unmat_open, right_in_n_unmat_close,
  input uwire [3:0] str [0:n-1] );

  if ( n == 1 ) begin

    assign left_out_n_unmat_close = str[0] == Char_Close;
    assign right_out_n_unmat_open = str[0] == Char_Open;

    assign str_marked[0] =
      str[0] == Char_Close && left_in_n_unmat_open ? Char_Close_Okay :
      str[0] == Char_Open && right_in_n_unmat_close ? Char_Open_Okay :
      str[0];
```

- ☐ Show the hardware that will be inferred for the base case ($n=1$) shown above.
- ☐ Show the hardware after optimization and ☐ for the default value of `wn`.
- ☐ In the optimized hardware do not show comparison units, instead show the individual gates performing the comparison, ☐ optimizing for constant values.

Problem 2: Appearing below are three variations on a module that will set its output to either the input value, or a maximum value if the input is larger. The module will always be instantiated with $w1 < wn$. All of them are functionally equivalent, but were synthesized to different costs (by Genus 23.12-s086.1 when similar code was used in the solution to Homework 3). Because they are functionally equivalent a perfect synthesis program would synthesize each to the same hardware (with equal costs).

```
module clamp_plan_a
  #( int w1 = 3, wn = 4 ) ( output uwire [w1-1:0] x, input uwire [wn-1:0] a );
  localparam logic [w1-1:0] nl_max = ~(w1)'(0); // Sequence of w1 1s.
  assign x = a <= nl_max ? a : nl_max;
endmodule
```

```
module clamp_plan_b
  #( int w1 = 3, wn = 4 ) ( output uwire [w1-1:0] x, input uwire [wn-1:0] a );
  localparam logic [w1-1:0] nl_max = ~(w1)'(0); // Sequence of w1 1s.
  assign x = a < nl_max ? a : nl_max;
endmodule
```

```
module clamp_plan_c
  #( int w1 = 3, wn = 4 ) ( output uwire [w1-1:0] x, input uwire [wn-1:0] a );
  localparam logic [w1-1:0] nl_max = ~(w1)'(0); // Sequence of w1 1s.
  assign x = !a[wn-1:w1] ? a : nl_max;
endmodule
```

☐ Show the optimized hardware for the low-cost version(s).

☐ Find the simple-model cost of each after optimization. The costs should be ☐ in terms of wn and $w1$.


☐ Find the simple-model delay of each after optimization. The delays should be ☐ in terms of wn and $w1$.

Problem 3: Appearing on the next page is a simplified solution to Homework 3, Problem 2. In this module the number of bits in the connections carrying parentheses counts is hardcoded to 8. Though the hardware is correct for $n < 256$ it is more costly and slower than it needs to be. But for this problem it's good enough.

Show the Homework 3, Problem 2 hardware that will be inferred for this module for $n > 1$ (the non-base case). That is, don't show the hardware computing `left_out_n_unmat_close` and `right_out_n_unmat_open`.

- ☐ Show the inferred hardware at one level for $n > 1$.
- ☐ Feel free to use abbreviations.
- ☐ Don't show the Homework 3 Problem 1 hardware (the last `always_comb`).
- ☐ Don't confuse elaboration-time computation with hardware.

For reference, the hardware for the Homework 3 Problem 1 part of this module is shown below.



```

module pmatch_mark_big #( int n = 5, wn = 8 )
( output logic [wn-1:0] left_out_n_unmat_close, right_out_n_unmat_open,
  output uwire [3:0] str_marked [0:n-1],
  input uwire [wn-1:0] left_in_n_unmat_open, right_in_n_unmat_close,
  input uwire [3:0] str [0:n-1] );

if ( n == 1 ) begin
  assign left_out_n_unmat_close = str[0] == Char_Close;
  assign right_out_n_unmat_open = str[0] == Char_Open;
  assign str_marked[0] =
    str[0] == Char_Close && left_in_n_unmat_open ? Char_Close_Okay :
    str[0] == Char_Open && right_in_n_unmat_close ? Char_Open_Okay : str[0];
end else begin

  localparam int n_left = n/2, n_right = n - n_left;
  localparam int wl = 8, wr = 8; // Note: this is wasteful.

  uwire [wl-1:0] lt_close, lt_open;
  uwire [wr-1:0] rt_close, rt_open;
  logic [wl-1:0] lt_matched_op, lt_matched_cl;
  logic [wr-1:0] rt_matched_op, rt_matched_cl;

  pmatch_mark_big #(n_left, wl) plt // Recursive Instantiation
  ( lt_close, lt_open, str_marked[0:n_left-1],
    lt_matched_cl, lt_matched_op, str[0:n_left-1] );
  pmatch_mark_big #(n_right, wr) prt // Recursive Instantiation
  ( rt_close, rt_open, str_marked[n_left:n-1],
    rt_matched_cl, rt_matched_op, str[n_left:n-1] );

  always_comb begin
    logic signed [wn-1:0] more_op, more_cl;

    lt_matched_cl = left_in_n_unmat_open;
    rt_matched_op = right_in_n_unmat_close;

    more_op = left_in_n_unmat_open - lt_close;
    rt_matched_cl = more_op < 0 ? lt_open : more_op + lt_open;

    more_cl = right_in_n_unmat_close - rt_open;
    lt_matched_op = more_cl < 0 ? rt_close : more_cl + rt_close;
  end

  always_comb begin // Same as Homework 3 Problem 1
    logic signed [wn-1:0] delta;
    delta = lt_open - rt_close;
    left_out_n_unmat_close = delta >= 0 ? lt_close : lt_close - delta;
    right_out_n_unmat_open = delta < 0 ? rt_open : rt_open + delta;
  end
end
endmodule

```